Double Pendulum Morin 6.19 5-25-16

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[10]:=
```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Purpose

I work through some basic calculations relevant to the double pendulum, motivated by Morin, *Introduction to Classical Mechanics: With Problems and Solutions*. Problem 6.19 is a nice “4-star” problem. Included are

1 derivation of the Lagrangian
2 generation of the Euler-Lagrange equation
3 linearization of the basic equations and analysis of special cases
4 eigenmode analysis
5 visualization and animation of the eigenmodes

Problem set up and geometry



Notational preparations

I want to use subscripts so I load the notation package and declare various entities to be symbols

In[12]:=
```
<< Notation`
```

In[73]:=
```
Symbolize[ x₁ ]; Symbolize[ y₁ ]; Symbolize[ x₂ ]; Symbolize[ y₂ ];
Symbolize[ m₁ ]; Symbolize[ ℓ₁ ]; Symbolize[ θ₁ ]; Symbolize[ m₂ ];
Symbolize[ ℓ₂ ];
Symbolize[ θ₂ ];
Symbolize[ δθ₁ ];
Symbolize[ δθ₂ ];
Symbolize[ e₁ ];
Symbolize[ e₂ ];
Symbolize[ C₁ ];
Symbolize[ C₂ ];
Symbolize[ C₃ ];
Symbolize[ C₄ ];
Symbolize[ μm ];
Symbolize[ A₁ ];
Symbolize[ A₂ ];
```

I Deriving the Lagrangian

I specify the end points of the two pendulums.

In[16]:=
```
w1[1] = Thread[{x₁[t], y₁[t]} → ℓ₁ {Sin[θ₁[t]], - Cos[θ₁[t]]}]
```

Out[16]=
```
{x₁[t] → ℓ₁ Sin[θ₁[t]], y₁[t] → -ℓ₁ Cos[θ₁[t]]}
```

In[17]:=
```
w1[2] =
 Thread[{x₂[t], y₂[t]} → {x₁[t], y₁[t]} + ℓ₂ {Sin[θ₂[t]], - Cos[θ₂[t]]}] /. w1[1]
```

Out[17]=
```
{x₂[t] → ℓ₁ Sin[θ₁[t]] + ℓ₂ Sin[θ₂[t]], y₂[t] → -ℓ₁ Cos[θ₁[t]] - ℓ₂ Cos[θ₂[t]]}
```

The potential energy is

In[18]:=
```
w1[3] = 𝒱[t] → m₁ g y₁[t] + m₂ g y₂[t]
```

Out[18]=
```
𝒱[t] → g m₁ y₁[t] + g m₂ y₂[t]
```

The kinetic energy is

In[19]:=
```
w1[4] = 𝒯[t] → (m1/2) (D[x1[t], t]^2 + D[y1[t], t]^2) + (m2/2) (D[x2[t], t]^2 + D[y2[t], t]^2)
```
Out[19]=
$$\mathcal{T}[t] \to \frac{1}{2} m_1 \left(x_1'[t]^2 + y_1'[t]^2\right) + \frac{1}{2} m_2 \left(x_2'[t]^2 + y_2'[t]^2\right)$$

The Lagrangian is

In[20]:=
```
w1[5] = ℒ[t] → 𝒯[t] - 𝒱[t] /. w1[3] /. w1[4]
```
Out[20]=
$$\mathcal{L}[t] \to -g\, m_1\, y_1[t] - g\, m_2\, y_2[t] + \frac{1}{2} m_1 \left(x_1'[t]^2 + y_1'[t]^2\right) + \frac{1}{2} m_2 \left(x_2'[t]^2 + y_2'[t]^2\right)$$

Using angular coordinates

In[21]:=
```
w1["final"] = w1[5] /. D[w1[1], t] /. D[w1[2], t] /. w1[1] /. w1[2] // Simplify
```
Out[21]=
$$\mathcal{L}[t] \to \frac{1}{2} \Big(2\, g \left(\ell_1 (m_1 + m_2) \, \mathrm{Cos}[\theta_1[t]] + \ell_2\, m_2\, \mathrm{Cos}[\theta_2[t]]\right) + \ell_1^2 (m_1 + m_2)\, \theta_1'[t]^2 + 2\, \ell_1\, \ell_2\, m_2\, \mathrm{Cos}[\theta_1[t] - \theta_2[t]]\, \theta_1'[t]\, \theta_2'[t] + \ell_2^2\, m_2\, \theta_2'[t]^2\Big)$$

2 Euler-Lagrange equations

The Euler-Lagrange equation for θ_1 is

In[22]:=
```
w2[2] =
 D[D[ℒ[t] /. w1["final"], θ1'[t]], t] == D[ℒ[t] /. w1["final"], θ1[t]] // ExpandAll
```
Out[22]=
$$-\ell_1 \ell_2 m_2 \mathrm{Sin}[\theta_1[t] - \theta_2[t]]\, \theta_1'[t]\, \theta_2'[t] + \ell_1 \ell_2 m_2 \mathrm{Sin}[\theta_1[t] - \theta_2[t]]\, \theta_2'[t]^2 + \ell_1^2 m_1 \theta_1''[t] + \ell_1^2 m_2 \theta_1''[t] + \ell_1 \ell_2 m_2 \mathrm{Cos}[\theta_1[t] - \theta_2[t]]\, \theta_2''[t] == -g\, \ell_1 m_1 \mathrm{Sin}[\theta_1[t]] - g\, \ell_1 m_2 \mathrm{Sin}[\theta_1[t]] - \ell_1 \ell_2 m_2 \mathrm{Sin}[\theta_1[t] - \theta_2[t]]\, \theta_1'[t]\, \theta_2'[t]$$

This can be simplified

In[23]:=
```
w2[3] = Collect[w2[2], {θ1''[t], θ2''[t]}];
w2[3] = Simplify[w2[3], Trig → True];
w2[3] = Collect[w2[3], { Sin[θ1[t]], g}]
```
Out[25]=
$$g\, \ell_1 (m_1 + m_2) \mathrm{Sin}[\theta_1[t]] + \ell_1 \left(\ell_2 m_2 \mathrm{Sin}[\theta_1[t] - \theta_2[t]]\, \theta_2'[t]^2 + \ell_1 (m_1 + m_2)\, \theta_1''[t] + \ell_2 m_2 \mathrm{Cos}[\theta_1[t] - \theta_2[t]]\, \theta_2''[t]\right) == 0$$

The Euler-Lagrange equation for θ_2 is

In[26]:=
```
w2[4] = D[D[ℒ[t] /. w1["final"], θ2'[t]], t] ==
    D[ℒ[t] /. w1["final"], θ2[t]] // ExpandAll;
w2[5] = Collect[w2[4], {θ1''[t], θ2''[t]}];
w2[5] = Simplify[w2[5], Trig → True]
```

Out[28]=
$$\ell_2 m_2 \left(g \operatorname{Sin}[\theta_2[t]] - \ell_1 \operatorname{Sin}[\theta_1[t] - \theta_2[t]]\, \theta_1'[t]^2 + \ell_1 \operatorname{Cos}[\theta_1[t] - \theta_2[t]]\, \theta_1''[t] + \ell_2 \theta_2''[t]\right) == 0$$

In[29]:=
```
w2["final"] = {w2[3], w2[5]}
```

Out[29]=
$$\{g \ell_1 (m_1 + m_2) \operatorname{Sin}[\theta_1[t]] + \ell_1 \left(\ell_2 m_2 \operatorname{Sin}[\theta_1[t] - \theta_2[t]]\, \theta_2'[t]^2 + \ell_1 (m_1 + m_2)\, \theta_1''[t] + \ell_2 m_2 \operatorname{Cos}[\theta_1[t] - \theta_2[t]]\, \theta_2''[t]\right) == 0,\ \ell_2 m_2 \left(g \operatorname{Sin}[\theta_2[t]] - \ell_1 \operatorname{Sin}[\theta_1[t] - \theta_2[t]]\, \theta_1'[t]^2 + \ell_1 \operatorname{Cos}[\theta_1[t] - \theta_2[t]]\, \theta_1''[t] + \ell_2 \theta_2''[t]\right) == 0\}$$

3 Limiting cases

Small oscillations approximation. Introduce the formal small quantity $\epsilon << 1$

In[31]:=
```
w3[1] = w2["final"] /. {θ1 → ((ϵ θ1[#]) &), θ2 → ((ϵ θ2[#]) &)}
```

Out[31]=
$$\{g \ell_1 (m_1 + m_2) \operatorname{Sin}[\epsilon \theta_1[t]] + \ell_1 \left(\ell_2 m_2 \epsilon^2 \operatorname{Sin}[\epsilon \theta_1[t] - \epsilon \theta_2[t]]\, \theta_2'[t]^2 + \ell_1 (m_1 + m_2)\, \epsilon\, \theta_1''[t] + \ell_2 m_2 \epsilon \operatorname{Cos}[\epsilon \theta_1[t] - \epsilon \theta_2[t]]\, \theta_2''[t]\right) == 0,\ \ell_2 m_2 \left(g \operatorname{Sin}[\epsilon \theta_2[t]] - \ell_1 \epsilon^2 \operatorname{Sin}[\epsilon \theta_1[t] - \epsilon \theta_2[t]]\, \theta_1'[t]^2 + \ell_1 \epsilon \operatorname{Cos}[\epsilon \theta_1[t] - \epsilon \theta_2[t]]\, \theta_1''[t] + \ell_2 \epsilon\, \theta_2''[t]\right) == 0\}$$

Expand and truncate at order ϵ

In[32]:=
```
w3[2] = Map[Normal@Series[#, {ϵ, 0, 2}] &, w3[1], {2}] // ExpandAll
```

Out[32]=
$$\{g \ell_1 m_1 \epsilon \theta_1[t] + g \ell_1 m_2 \epsilon \theta_1[t] + \ell_1^2 m_1 \epsilon \theta_1''[t] + \ell_1^2 m_2 \epsilon \theta_1''[t] + \ell_1 \ell_2 m_2 \epsilon \theta_2''[t] == 0,\ g \ell_2 m_2 \epsilon \theta_2[t] + \ell_1 \ell_2 m_2 \epsilon \theta_1''[t] + \ell_2^2 m_2 \epsilon \theta_2''[t] == 0\}$$

In[33]:=
```
w3[3] = w3[2] /. ϵ^(n_/;n>1) → 0 /. ϵ → 1
```

Out[33]=
$$\{g \ell_1 m_1 \theta_1[t] + g \ell_1 m_2 \theta_1[t] + \ell_1^2 m_1 \theta_1''[t] + \ell_1^2 m_2 \theta_1''[t] + \ell_1 \ell_2 m_2 \theta_2''[t] == 0,\ g \ell_2 m_2 \theta_2[t] + \ell_1 \ell_2 m_2 \theta_1''[t] + \ell_2^2 m_2 \theta_2''[t] == 0\}$$

In[34]:=
```
w3["final"] = w3[3]
```

Out[34]=
$$\{g \ell_1 m_1 \theta_1[t] + g \ell_1 m_2 \theta_1[t] + \ell_1^2 m_1 \theta_1''[t] + \ell_1^2 m_2 \theta_1''[t] + \ell_1 \ell_2 m_2 \theta_2''[t] == 0,\ g \ell_2 m_2 \theta_2[t] + \ell_1 \ell_2 m_2 \theta_1''[t] + \ell_2^2 m_2 \theta_2''[t] == 0\}$$

A Special case: Equal length pendulums

In[36]:=
```
w3A[1] = w3[3] /. ℓ2 → ℓ1 /. ℓ1 → ℓ
```

Out[36]=

$$\{g\,\ell\,m_1\,\Theta_1[t] + g\,\ell\,m_2\,\Theta_1[t] + \ell^2\,m_1\,\Theta_1''[t] + \ell^2\,m_2\,\Theta_1''[t] + \ell^2\,m_2\,\Theta_2''[t] == 0,\; g\,\ell\,m_2\,\Theta_2[t] + \ell^2\,m_2\,\Theta_1''[t] + \ell^2\,m_2\,\Theta_2''[t] == 0\}$$

Special case: $m_2 << m_1$

In[37]:=
```
Module[{θ1 = π/16, θ2 = -π/20, μm = 0.9, μℓ = 0.5, lab},
 lab = "Example m2 << m1\nBehaves like pendulum with length ℓ1";
 ShowPendulums[{θ1, θ2}, μℓ, μm, lab]]
```

Out[37]=



Introduce a parameter for the mass ratio

In[38]:=
```
w3A[2] = w3A[1] /. m2 → μ m1
```

Out[38]=

$$\{g\,\ell\,m_1\,\Theta_1[t] + g\,\ell\,m_1\,\mu\,\Theta_1[t] + \ell^2\,m_1\,\Theta_1''[t] + \ell^2\,m_1\,\mu\,\Theta_1''[t] + \ell^2\,m_1\,\mu\,\Theta_2''[t] == 0,\; g\,\ell\,m_1\,\mu\,\Theta_2[t] + \ell^2\,m_1\,\mu\,\Theta_1''[t] + \ell^2\,m_1\,\mu\,\Theta_2''[t] == 0\}$$

In[39]:=
```
w3A[3] = Map[(#/(m1 ℓ^2)) &, w3A[2], {2}] // Expand
```

Out[39]=

$$\left\{\frac{g\,\Theta_1[t]}{\ell} + \frac{g\,\mu\,\Theta_1[t]}{\ell} + \Theta_1''[t] + \mu\,\Theta_1''[t] + \mu\,\Theta_2''[t] == 0,\; \frac{g\,\mu\,\Theta_2[t]}{\ell} + \mu\,\Theta_1''[t] + \mu\,\Theta_2''[t] == 0\right\}$$

In the limit $\mu \to 0$, this case reduces to that of a simple pendulum of with mass m_1 and length l_1. The second pendulum is negligible.

In[40]:=
```
w3A[4] = w3A[3] /. μ → 0
```

Out[40]=
$$\left\{\frac{g\,\theta_1[t]}{\ell} + \theta_1''[t] == 0,\ \text{True}\right\}$$

Special case: $m_1 << m_2$

In[41]:=
```
Module[{θ1 = π/16, θ2 = -π/20, μm = 0.1, μℓ = 0.5, lab},
 lab = "Example m1 << m2\nBehaves like pendulum with length ℓ1+ℓ2";
 ShowPendulums[{θ1, θ2}, μℓ, μm, lab]]
```

Out[41]=



In[42]:=
```
w3A[5] = w3A[1] /. m1 → η m2
```

Out[42]=
$$\{g\,\ell\,m_2\,\theta_1[t] + g\,\ell\,m_2\,\eta\,\theta_1[t] + \ell^2\,m_2\,\theta_1''[t] + \ell^2\,m_2\,\eta\,\theta_1''[t] + \ell^2\,m_2\,\theta_2''[t] == 0,$$
$$g\,\ell\,m_2\,\theta_2[t] + \ell^2\,m_2\,\theta_1''[t] + \ell^2\,m_2\,\theta_2''[t] == 0\}$$

In[43]:=
```
w3A[6] = Map[(#/(m2 ℓ^2)) &, w3A[5] , {2}] // Expand
```

Out[43]=
$$\left\{\frac{g\,\theta_1[t]}{\ell} + \frac{g\,\eta\,\theta_1[t]}{\ell} + \theta_1''[t] + \eta\,\theta_1''[t] + \theta_2''[t] == 0,\ \frac{g\,\theta_2[t]}{\ell} + \theta_1''[t] + \theta_2''[t] == 0\right\}$$

In[44]:=
```
w3A[7] = w3A[6] /. η → 0
```

Out[44]=
$$\left\{\frac{g\,\theta_1[t]}{\ell} + \theta_1''[t] + \theta_2''[t] == 0,\ \frac{g\,\theta_2[t]}{\ell} + \theta_1''[t] + \theta_2''[t] == 0\right\}$$

In this case, the motion reduces to that of a pendulum with length 2 l.

In[45]:=
```
w3A[8] = w3A[7] /. θ1 → θ2;
w3A[8] = Map[(#/2) &, w3A[8], {2}] // Expand
```

Out[46]=

$$\left\{\frac{g\,\theta_2[t]}{2\,\ell}+\theta_2''[t]==0,\ \frac{g\,\theta_2[t]}{2\,\ell}+\theta_2''[t]==0\right\}$$

3B Special case: Equal masses

In[47]:=
```
w3B[1] = w3["final"] /. m2 → m1 /. m1 → m
```

Out[47]=

$$\left\{2\,g\,\ell_1\,m\,\theta_1[t]+2\,\ell_1^2\,m\,\theta_1''[t]+\ell_1\,\ell_2\,m\,\theta_2''[t]==0,\ g\,\ell_2\,m\,\theta_2[t]+\ell_1\,\ell_2\,m\,\theta_1''[t]+\ell_2^2\,m\,\theta_2''[t]==0\right\}$$

Special case: $l_2 << l_1$

In[48]:=
```
Module[{θ1 = π/16, θ2 = -π/20, μm = 0.5, μℓ = 0.9, lab},
 lab = "Example ℓ2 << ℓ1\nBehaves like pendulum with length ℓ1";
 ShowPendulums[{θ1, θ2}, μℓ, μm, lab]]
```

Out[48]=



In[49]:=
```
w3B[2] = w3B[1] /. ℓ2 → μ ℓ1
```

Out[49]=

$$\left\{2\,g\,\ell_1\,m\,\theta_1[t]+2\,\ell_1^2\,m\,\theta_1''[t]+\ell_1^2\,m\,\mu\,\theta_2''[t]==0,\right.$$
$$\left.g\,\ell_1\,m\,\mu\,\theta_2[t]+\ell_1^2\,m\,\mu\,\theta_1''[t]+\ell_1^2\,m\,\mu^2\,\theta_2''[t]==0\right\}$$

In[50]:=
```
w3B[3] = Map[(#/(2 m ℓ1^2)) &, w3B[2], {2}] // Expand
```

Out[50]=
$$\left\{\frac{g\,\theta_1[t]}{\ell_1}+\theta_1''[t]+\frac{1}{2}\mu\,\theta_2''[t]==0,\ \frac{g\,\mu\,\theta_2[t]}{2\,\ell_1}+\frac{1}{2}\mu\,\theta_1''[t]+\frac{1}{2}\mu^2\,\theta_2''[t]==0\right\}$$

In[51]:=
```
w3B[4] = w3B[3] /. μ → 0
```

Out[51]=
$$\left\{\frac{g\,\theta_1[t]}{\ell_1}+\theta_1''[t]==0,\ \text{True}\right\}$$

and the motion is that of a simple pendulum with length l_1

Special case: $l_1 << l_2$

In[52]:=
```
Module[{θ1 = π/16, θ2 = -π/20, μm = 0.5, μℓ = 0.1, lab},
 lab = "Example ℓ1 << ℓ2\nBehaves like pendulum with length ℓ2";
 ShowPendulums[{θ1, θ2}, μℓ, μm, lab]]
```

Out[52]=



In[53]:=
```
w3B[5] = w3B[1] /. ℓ1 → η ℓ2
```

Out[53]=
$$\{2\,g\,\ell_2\,m\,\eta\,\theta_1[t]+2\,\ell_2^2\,m\,\eta^2\,\theta_1''[t]+\ell_2^2\,m\,\eta\,\theta_2''[t]==0,$$
$$g\,\ell_2\,m\,\theta_2[t]+\ell_2^2\,m\,\eta\,\theta_1''[t]+\ell_2^2\,m\,\theta_2''[t]==0\}$$

In[54]:=
```
w3B[6] = Map[(#/(m ℓ2^2)) &, w3B[5], {2}] // Expand
```

Out[54]=
$$\left\{\frac{2\,g\,\eta\,\theta_1[t]}{\ell_2}+2\,\eta^2\,\theta_1''[t]+\eta\,\theta_2''[t]==0,\ \frac{g\,\theta_2[t]}{\ell_2}+\eta\,\theta_1''[t]+\theta_2''[t]==0\right\}$$

In[55]:=
```
w3B[7] = w3B[6] /. η → 0
```

Out[55]=
$$\left\{\text{True}, \frac{g\,\theta_2[t]}{\ell_2} + \theta_2''[t] == 0\right\}$$

Again, the simple pendulum but with length l_2.

4 Eigenmodes, equal length pendulums

To illustrate eigenmodes I consider equal length pendulums. The equal mass case is analogous.

In[56]:=
```
w4[1] = w3["final"] /. {ℓ1 → ℓ, ℓ2 → ℓ}
```

Out[56]=
$$\{g\,\ell\,m_1\,\theta_1[t] + g\,\ell\,m_2\,\theta_1[t] + \ell^2\,m_1\,\theta_1''[t] + \ell^2\,m_2\,\theta_1''[t] + \ell^2\,m_2\,\theta_2''[t] == 0,\; g\,\ell\,m_2\,\theta_2[t] + \ell^2\,m_2\,\theta_1''[t] + \ell^2\,m_2\,\theta_2''[t] == 0\}$$

Determine the equilibrium state

In[57]:=
```
w4[2] = w4[1] /. {θ1'[t] → 0, θ1''[t] → 0} /. {θ2'[t] → 0, θ2''[t] → 0}
```

Out[57]=
$$\{g\,\ell\,m_1\,\theta_1[t] + g\,\ell\,m_2\,\theta_1[t] == 0,\; g\,\ell\,m_2\,\theta_2[t] == 0\}$$

As expected the equilibrium solutions correspond to the pendulums hanging vertically.

In[58]:=
```
w4[3] = Solve[w4[2], {θ1[t], θ2[t]}][[1]]
```

Out[58]=
$$\{\theta_1[t] \to 0,\; \theta_2[t] \to 0\}$$

The generation of perturbations about the equilibrium is trivial in this case.

In[59]:=
```
w4[4] = w4[1] /. θ1 → ((ϵ δθ1[#] &)) /. θ2 → ((ϵ δθ2[#] &))
```

Out[59]=
$$\{g\,\ell\,m_1\,\epsilon\,\delta\theta_1[t] + g\,\ell\,m_2\,\epsilon\,\delta\theta_1[t] + \ell^2\,m_1\,\epsilon\,\delta\theta_1''[t] + \ell^2\,m_2\,\epsilon\,\delta\theta_1''[t] + \ell^2\,m_2\,\epsilon\,\delta\theta_2''[t] == 0,\; g\,\ell\,m_2\,\epsilon\,\delta\theta_2[t] + \ell^2\,m_2\,\epsilon\,\delta\theta_1''[t] + \ell^2\,m_2\,\epsilon\,\delta\theta_2''[t] == 0\}$$

In[60]:=
```
w4[5] = w4[4] /. ϵ^(n_/;n>1) → 0 /. ϵ → 1
```

Out[60]=
$$\{g\,\ell\,m_1\,\delta\theta_1[t] + g\,\ell\,m_2\,\delta\theta_1[t] + \ell^2\,m_1\,\delta\theta_1''[t] + \ell^2\,m_2\,\delta\theta_1''[t] + \ell^2\,m_2\,\delta\theta_2''[t] == 0,\; g\,\ell\,m_2\,\delta\theta_2[t] + \ell^2\,m_2\,\delta\theta_1''[t] + \ell^2\,m_2\,\delta\theta_2''[t] == 0\}$$

In[61]:=
```
w4[6] = Map[(#/(m1 ℓ^2)) &, w4[5], {2}] // Expand
```

Out[61]=
$$\left\{\frac{g\,\delta\theta_1[t]}{\ell} + \frac{g\,m_2\,\delta\theta_1[t]}{\ell\,m_1} + \delta\theta_1''[t] + \frac{m_2\,\delta\theta_1''[t]}{m_1} + \frac{m_2\,\delta\theta_2''[t]}{m_1} == 0,\; \frac{g\,m_2\,\delta\theta_2[t]}{\ell\,m_1} + \frac{m_2\,\delta\theta_1''[t]}{m_1} + \frac{m_2\,\delta\theta_2''[t]}{m_1} == 0\right\}$$

It is convenient to adopt a dimensionless form

In[62]:=
```
def[Ω] = Ω^2 == g / ℓ
```

Out[62]=
$$\Omega^2 == \frac{g}{\ell}$$

In[63]:=
```
w4[7] = w4[6] /. Sol[def[Ω], g]
```

Out[63]=
$$\left\{\Omega^2\,\delta\Theta_1[t] + \frac{m_2\,\Omega^2\,\delta\Theta_1[t]}{m_1} + \delta\Theta_1''[t] + \frac{m_2\,\delta\Theta_1''[t]}{m_1} + \frac{m_2\,\delta\Theta_2''[t]}{m_1} == 0,\; \frac{m_2\,\Omega^2\,\delta\Theta_2[t]}{m_1} + \frac{m_2\,\delta\Theta_1''[t]}{m_1} + \frac{m_2\,\delta\Theta_2''[t]}{m_1} == 0\right\}$$

In[64]:=
```
w4[8] = {w4[7][[1, 1]], w4[7][[2, 1]] m1 / m2}  // ExpandAll
```

Out[64]=
$$\left\{\Omega^2\,\delta\Theta_1[t] + \frac{m_2\,\Omega^2\,\delta\Theta_1[t]}{m_1} + \delta\Theta_1''[t] + \frac{m_2\,\delta\Theta_1''[t]}{m_1} + \frac{m_2\,\delta\Theta_2''[t]}{m_1},\; \Omega^2\,\delta\Theta_2[t] + \delta\Theta_1''[t] + \delta\Theta_2''[t]\right\}$$

Construct the eigenvalue equations

In[65]:=
```
w4[9] = w4[8] /. {δΘ1 → ((A Exp[I ω #]) &), δΘ2 → ((B Exp[I ω #]) &)} /. t → 0
```

Out[65]=
$$\left\{-A\,\omega^2 - \frac{A\,m_2\,\omega^2}{m_1} - \frac{B\,m_2\,\omega^2}{m_1} + A\,\Omega^2 + \frac{A\,m_2\,\Omega^2}{m_1},\; -A\,\omega^2 - B\,\omega^2 + B\,\Omega^2\right\}$$

In[66]:=
```
w4[10] = {{Coefficient[w4[9][[1]], A], Coefficient[w4[9][[1]], B]},
  {Coefficient[w4[9][[2]], A], Coefficient[w4[9][[2]], B]}}
```

Out[66]=
$$\left\{\left\{-\omega^2 - \frac{m_2\,\omega^2}{m_1} + \Omega^2 + \frac{m_2\,\Omega^2}{m_1},\; -\frac{m_2\,\omega^2}{m_1}\right\},\; \left\{-\omega^2,\; -\omega^2 + \Omega^2\right\}\right\}$$

In[67]:=
```
w4[10] // MatrixForm
```

Out[67]//MatrixForm=
$$\begin{pmatrix} -\omega^2 - \frac{m_2\,\omega^2}{m_1} + \Omega^2 + \frac{m_2\,\Omega^2}{m_1} & -\frac{m_2\,\omega^2}{m_1} \\ -\omega^2 & -\omega^2 + \Omega^2 \end{pmatrix}$$

The eigenmode equation is

In[68]:=
```
w4[11] = Det[w4[10]] == 0
```

Out[68]=
$$\omega^4 - 2\,\omega^2\,\Omega^2 - \frac{2\,m_2\,\omega^2\,\Omega^2}{m_1} + \Omega^4 + \frac{m_2\,\Omega^4}{m_1} == 0$$

In[69]:=
```
w4[12] = Solve[w4[11], ω];
w4[12] = Simplify[#, Assumptions → {Ω ∈ Reals, Ω > 0}] & /@ w4[12]
```

Out[70]=

$$\left\{\left\{\omega \to -\sqrt{\left(\frac{1}{m_1}\left(m_1+m_2-\sqrt{m_2\,(m_1+m_2)}\right)\right)}\,\Omega\right\},\ \left\{\omega \to \sqrt{\frac{m_1+m_2-\sqrt{m_2\,(m_1+m_2)}}{m_1}}\,\Omega\right\},\ \left\{\omega \to -\sqrt{\left(\frac{1}{m_1}\left(m_1+m_2+\sqrt{m_2\,(m_1+m_2)}\right)\right)}\,\Omega\right\},\ \left\{\omega \to \sqrt{\frac{m_1+m_2+\sqrt{m_2\,(m_1+m_2)}}{m_1}}\,\Omega\right\}\right\}$$

These expressions agree with the eigenvalues given by Morin. Now for the eigenmodes

In[71]:=
```
w4[13] = {w4[12]〚2, 1〛, w4[12]〚4, 1〛}
```

Out[71]=

$$\left\{\omega \to \sqrt{\frac{m_1+m_2-\sqrt{m_2\,(m_1+m_2)}}{m_1}}\,\Omega,\ \omega \to \sqrt{\frac{m_1+m_2+\sqrt{m_2\,(m_1+m_2)}}{m_1}}\,\Omega\right\}$$

The eigenvalues are

In[72]:=
```
w4[14] = {w4[13]〚1〛 /. ω → ω1, w4[13]〚2〛 /. ω → ω2}
```

Out[72]=

$$\left\{\omega_1 \to \sqrt{\frac{m_1+m_2-\sqrt{m_2\,(m_1+m_2)}}{m_1}}\,\Omega,\ \omega_2 \to \sqrt{\frac{m_1+m_2+\sqrt{m_2\,(m_1+m_2)}}{m_1}}\,\Omega\right\}$$

I introduce some simplifying notation

In[76]:=
```
w4[15] = w4[14] /. m1 → μm m2;
w4[15] = Simplify[#, {m2 ∈ Reals, m2 > 0}] & /@ w4[15]
```

Out[77]=

$$\left\{\omega_1 \to \Omega\sqrt{\frac{1+\mu_m-\sqrt{1+\mu_m}}{\mu_m}},\ \omega_2 \to \Omega\sqrt{\frac{1+\mu_m+\sqrt{1+\mu_m}}{\mu_m}}\right\}$$

To determine the eigenmodes, Choose either of the equations

In[78]:=
```
w4[9] /. m1 → μm m2
```

Out[78]=

$$\left\{-A\,\omega^2 + A\,\Omega^2 - \frac{A\,\omega^2}{\mu_m} - \frac{B\,\omega^2}{\mu_m} + \frac{A\,\Omega^2}{\mu_m},\ -A\,\omega^2 - B\,\omega^2 + B\,\Omega^2\right\}$$

and set B =1

In[79]:=
```
w4[16] = Sol[w4[9]〚2〛 == 0 /. B → 1, A]
```

Out[79]=

$$A \to \frac{-\omega^2+\Omega^2}{\omega^2}$$

The two possibilities for A are

In[80]:=
```
w4[17] = {w4[16] /. ω → ω₁ /. A → A₁, w4[16] /. ω → ω₂ /. A → A₂}
```

Out[80]=
$$\left\{A_1 \to \frac{\Omega^2 - \omega_1^2}{\omega_1^2},\ A_2 \to \frac{\Omega^2 - \omega_2^2}{\omega_2^2}\right\}$$

Now to construct specific forms for the eigenmodes

In[81]:=
```
w4[18] = {e₁ → {A₁, 1} , e₂ → {A₂, 1}} /. w4[17]
```

Out[81]=
$$\left\{e_1 \to \left\{\frac{\Omega^2 - \omega_1^2}{\omega_1^2},\ 1\right\},\ e_2 \to \left\{\frac{\Omega^2 - \omega_2^2}{\omega_2^2},\ 1\right\}\right\}$$

Now to express the solution in terms of eigenmodes

In[82]:=
```
w4[19] =
 {δθ₁[t] == e₁.{C₁ Exp[I ω₁ t] + C₂ Exp[-I ω₁ t] , C₃ Exp[I ω₂ t] + C₄ Exp[-I ω₂ t]},
  δθ₂[t] == e₂.{C₁ Exp[I ω₁ t] + C₂  Exp[-I ω₁ t] , C₃ Exp[I ω₂ t] + C₄ Exp[-I ω₂ t]}}
```

Out[82]=
$$\left\{\delta\theta_1[t] == e_1.\left\{C_2\, e^{-i t \omega_1} + C_1\, e^{i t \omega_1},\ C_4\, e^{-i t \omega_2} + C_3\, e^{i t \omega_2}\right\},\right.$$
$$\left.\delta\theta_2[t] == e_2.\left\{C_2\, e^{-i t \omega_1} + C_1\, e^{i t \omega_1},\ C_4\, e^{-i t \omega_2} + C_3\, e^{i t \omega_2}\right\}\right\}$$

The eigenmodes must be real

In[83]:=
```
w4[20] = w4[19] /. C₂ → Conjugate[C₁] /. C₄ → Conjugate[C₃]
```

Out[83]=
$$\left\{\delta\theta_1[t] == e_1.\left\{C_1\, e^{i t \omega_1} + e^{-i t \omega_1}\,\mathrm{Conjugate}[C_1],\ C_3\, e^{i t \omega_2} + e^{-i t \omega_2}\,\mathrm{Conjugate}[C_3]\right\},\right.$$
$$\left.\delta\theta_2[t] == e_2.\left\{C_1\, e^{i t \omega_1} + e^{-i t \omega_1}\,\mathrm{Conjugate}[C_1],\ C_3\, e^{i t \omega_2} + e^{-i t \omega_2}\,\mathrm{Conjugate}[C_3]\right\}\right\}$$

```
Symbolize[ R₁ ]; Symbolize[ ϕ₁ ]; Symbolize[ R₂ ]; Symbolize[ ϕ₂ ];
```

In[84]:=
```
w4[21] = w4[20] /. {C₁ → R₁ Exp[I ϕ₁] / 2, Conjugate[C₁] →  R₁ Exp[-I ϕ₁] / 2} /.
   {C₃ → R₂ Exp[I ϕ₂] / 2, Conjugate[C₃] →  R₂ Exp[-I ϕ₂] / 2}
```

Out[84]=
$$\left\{\delta\theta_1[t] == e_1.\left\{\frac{1}{2}\, e^{-i \phi_1 - i t \omega_1} R_1 + \frac{1}{2}\, e^{i \phi_1 + i t \omega_1} R_1,\ \frac{1}{2}\, e^{-i \phi_2 - i t \omega_2} R_2 + \frac{1}{2}\, e^{i \phi_2 + i t \omega_2} R_2\right\},\right.$$
$$\left.\delta\theta_2[t] == e_2.\left\{\frac{1}{2}\, e^{-i \phi_1 - i t \omega_1} R_1 + \frac{1}{2}\, e^{i \phi_1 + i t \omega_1} R_1,\ \frac{1}{2}\, e^{-i \phi_2 - i t \omega_2} R_2 + \frac{1}{2}\, e^{i \phi_2 + i t \omega_2} R_2\right\}\right\}$$

In[85]:=
```
w4[22] = Map[ExpToTrig, w4[21], {3}]
```

Out[85]=
$$\{\delta\theta_1[t] == e_1.\{\mathrm{Cos}[\phi_1 + t\,\omega_1]\, R_1,\ \mathrm{Cos}[\phi_2 + t\,\omega_2]\, R_2\},$$
$$\delta\theta_2[t] == e_2.\{\mathrm{Cos}[\phi_1 + t\,\omega_1]\, R_1,\ \mathrm{Cos}[\phi_2 + t\,\omega_2]\, R_2\}\}$$

In[86]:=
```
w4[23] = w4[22] /. {e1 → {A1, 1}, e2 → {A2, 1}}
```

Out[86]=

$$\{\delta\Theta_1[t] == A_1 \text{Cos}[\phi_1 + t\,\omega_1]\, R_1 + \text{Cos}[\phi_2 + t\,\omega_2]\, R_2,\ \delta\Theta_2[t] == A_2 \text{Cos}[\phi_1 + t\,\omega_1]\, R_1 + \text{Cos}[\phi_2 + t\,\omega_2]\, R_2\}$$

Summarizing, the eigenvalues are

In[87]:=
```
w4[15]
```

Out[87]=

$$\left\{\omega_1 \to \Omega \sqrt{\frac{1 + \mu_m - \sqrt{1 + \mu_m}}{\mu_m}},\ \omega_2 \to \Omega \sqrt{\frac{1 + \mu_m + \sqrt{1 + \mu_m}}{\mu_m}}\right\}$$

and the corresponding eigenmodes are

In[88]:=
```
w4[18]
```

Out[88]=

$$\left\{e_1 \to \left\{\frac{\Omega^2 - \omega_1^2}{\omega_1^2}, 1\right\},\ e_2 \to \left\{\frac{\Omega^2 - \omega_2^2}{\omega_2^2}, 1\right\}\right\}$$

The constants R and ϕ are determined from initial conditions.

5 Visualization and animation of eigenmodes

In[89]:=
```
w5[1] = {emode1 == {A1 Cos[t ω1], Cos[t ω1]}, emode2 == {A2 Cos[t ω2], Cos[t ω2]}}
```

Out[89]=

$$\{\text{emode1} == \{A_1 \text{Cos}[t\,\omega_1], \text{Cos}[t\,\omega_1]\},\ \text{emode2} == \{A_2 \text{Cos}[t\,\omega_2], \text{Cos}[t\,\omega_2]\}\}$$

In[90]:=
```
w5[2] = w5[1] /. t → τ / Ω /. {ω1 → ω1, ω2 → ω2, A1 → A1, A2 → A2}
```

Out[90]=

$$\left\{\text{emode1} == \left\{\text{A1 Cos}\left[\frac{\tau\,\omega 1}{\Omega}\right], \text{Cos}\left[\frac{\tau\,\omega 1}{\Omega}\right]\right\},\ \text{emode2} == \left\{\text{A2 Cos}\left[\frac{\tau\,\omega 2}{\Omega}\right], \text{Cos}\left[\frac{\tau\,\omega 2}{\Omega}\right]\right\}\right\}$$

In[91]:=
```
w4[15] /. μm → μm
```

Out[91]=

$$\left\{\omega_1 \to \sqrt{\frac{1 + \mu m - \sqrt{1 + \mu m}}{\mu m}}\ \Omega,\ \omega_2 \to \sqrt{\frac{1 + \mu m + \sqrt{1 + \mu m}}{\mu m}}\ \Omega\right\}$$

In[92]:=
```
w4[18] /. {ω1 → ω1, ω2 → ω2}
```

Out[92]=

$$\left\{e_1 \to \left\{\frac{\Omega^2 - \omega 1^2}{\omega 1^2}, 1\right\},\ e_2 \to \left\{\frac{\Omega^2 - \omega 2^2}{\omega 2^2}, 1\right\}\right\}$$

Time dependence of eigenmodes

In[93]:=

```
Module[{g = 9.8, ℓ = 1, μm = 1, image = {500, 250}, Ω, ω1, ω2, A1, A2, lab, G},
 Ω = Sqrt[g / ℓ];
 {ω1, ω2} = {Sqrt[(1 + μm - Sqrt[1 + μm])/μm] Ω, Sqrt[(1 + μm + Sqrt[1 + μm])/μm] Ω};
 {A1, A2} = {(Ω^2 - ω1^2)/ω1^2, (Ω^2 - ω2^2)/ω2^2};
 lab = StringForm["μm = ``   ω₁ = `` = ``",
   NF2@μm, TraditionalForm["Sqrt[(1/μm (1 + μm - Sqrt[1 + μm]))] Ω"], NF2[ω1]];
 G[1] = Plot[{A1 Cos[τ ω1/Ω], Cos[τ ω1/Ω]}, {τ, 0, 10},
   PlotStyle → {Black, DASHED},
   PlotLabel → Stl@lab,
   AxesLabel → {Stl["τ"], Stl["A₁(black), B(dashed)"]}, ImageSize → image];
 lab = StringForm["μm = ``   ω₂ = `` = ``", NF2@μm,
   TraditionalForm["Sqrt[(1/μm (1 + μm + Sqrt[1 + μm]))] Ω"], NF2[ω2]];
 G[2] = Plot[{A2 Cos[τ ω2/Ω], Cos[τ ω2/Ω]}, {τ, 0, 10},
   PlotStyle → {Black, DASHED},
   PlotLabel → Stl@lab,
   AxesLabel → {Stl["τ"], Stl["A₂(black), B(dashed)"]}, ImageSize → image];
 Grid[{{G[1]}, {G[2]}}]]
```

Out[93]=


μm = 1.00 ω1 = √((1+μm−√(1+μm))/μm) Ω = 2.40
A1(black), B(dashed)
1.0
0.5
−0.5
−1.0
2
4
6
8
10
τ


μm = 1.00 ω2 = √((1+μm+√(1+μm))/μm) Ω = 5.78
A2(black), B(dashed)
1.0
0.5
−0.5
−1.0
2
4
6
8
10
τ

Animation of eigenmodes

In[94]:=

```
Module[{g = 9.8, ℓ = 1, μm = .2, μlPlot,
  μmPlot, Ω, ω1, ω2, A1, A2, vals1, vals2, lab, frames},
 Ω = Sqrt[g / ℓ];
 {ω1, ω2} = {Sqrt[(1 + μm - Sqrt[1 + μm]) / μm] Ω, Sqrt[(1 + μm + Sqrt[1 + μm]) / μm] Ω};
 {A1, A2} = {(Ω^2 - ω1^2) / ω1^2, (Ω^2 - ω2^2) / ω2^2};
 vals1 = Table[0.1 {A1 Cos[τ ω1 / Ω], Cos[τ ω1 / Ω]}, {τ, 0, 10, 0.1}];
 vals2 = Table[0.1 {A2 Cos[τ ω2 / Ω], Cos[τ ω2 / Ω]}, {τ, 0, 10, 0.1}];
 μlPlot = 0.5;
 μmPlot = μm / (1 + μm);
 lab[1] = StringForm["Mode 1 (symmetric)\nμ_m = m_1/m_2 = ``", NF2@μm];
 lab[2] = StringForm["Mode 2 (antisym)\nμ_m = m_1/m_2 = ``", NF2@μm];
 frames =
  Table[Grid[{{ShowEigenmodes[vals1[[i]], μlPlot, μmPlot, lab[1], {300, 250}],
      ShowEigenmodes[vals2[[i]], μlPlot, μmPlot, lab[2], {300, 250}]}}], {i, 1, 100}];
 ListAnimate[frames, 10]]
```

Out[94]=



Graphics

In[95]:=

```
InitialDiagram[{π / 4, -π/8}, 0.4, 0.4, "Double Pendulum"]
```

Out[95]=



In[4]:=

```
Clear[InitialDiagram];
InitialDiagram[{θ1_, θ2_}, μℓ_, μm_, lab_] :=
 Module[{ℓ = 1, m = 1, g = 9.8, offset = 0.075,
   pSize = 0.10, x1, x2, y1, y2, ℓ1, ℓ2, pSize1, pSize2, O, P1, P2,
   pendula, support, vertical, θarc, xAxis, zAxis, axes, PtoC, GVector},
  PtoC[r_, θ_] := r {Sin[θ], Cos[θ]};
  GVector[tail_, tip_, label_, posLabel_, size_, color_] :=
   {color, Arrowheads[size], Arrow[{tail, tip}], Text[Stl[label], posLabel]};

  support = {BLACK, Line[{{-0.5, 0}, {0.5, 0}}]};
  vertical = {GrayLevel[0.75], Dashing[{0.01, 0.01}], Line[{{0, 0}, {0, -1.1}}]};
  {ℓ1, ℓ2} = {μℓ , 1 - μℓ} ℓ;
  x1 = ℓ1 Sin[θ1];
  y1 = -ℓ1 Cos[θ1];
  x2 = ℓ1 Sin[θ1] + ℓ2 Sin[θ2];
  y2 = -ℓ1 Cos[θ1] - ℓ2 Cos[θ2];
  {pSize1, pSize2} = {μm, 1 - μm} pSize;
  {O, P1, P2} = {{0, 0}, {x1, y1}, {x2, y2}};
  axes = With[{origin = {0.5, -0.5}},
    xAxis = GVector[origin, origin + {0.1, 0}, "x", origin + {0.15, 0}, Small, Gray];
    zAxis = GVector[origin, origin + {0, 0.1}, "y", origin + {0, 0.15}, Small, Gray];
    {xAxis, zAxis}];
  θarc = {GrayLevel[0.7], Arrowheads[Small] ,
    Arrow@Table[PtoC[ℓ1/3, θ], {θ, π, π - θ1, - θ1/64}],
    {Black, Text[Stl["θ₁"], PtoC[1.25 ℓ1/3, Mean[{π, π - θ1}]]]}};
  pendula =
   {Thick, {Brown, Line[{O, P1}], {Black, Text[Stl["ℓ₁"], (O + P1)/2 + {1, 0} offset],
      PointSize@pSize1, Point[P1], Text[Stl["m₁"], P1 + {1, 0} offset]}},
    {Yellow, Line[{P1, P2}], {Black, Text[Stl["ℓ₂"], (P1 + P2)/2 + {1, 0} offset],
      PointSize@pSize2, Point[P2], Text[Stl["m₂"], P2 + {1, 0} offset]}}};

  Graphics[{support, vertical, pendula, θarc, axes},
   PlotLabel → Style[lab, 10, Bold], AspectRatio → 1, ImageSize → {400, 350}]]
```

In[96]:=

```
ShowPendulums[{π / 4, -π/8}, 0.4, 0.4, "Example State of Double Pendulum"]
```

Out[96]=



In[6]:=
```
Clear[ShowPendulums];
ShowPendulums[{θ1_, θ2_}, μℓ_, μm_, lab_] :=
 Module[{ℓ = 1, m = 1, g = 9.8, offset = 0.075,
   pSize = 0.10, x1, x2, y1, y2, ℓ1, ℓ2, m1, m2, pSize1, pSize2, O,
   P1, P2, pendula, support, vertical, insert, inset, PtoC, GVector},
  PtoC[r_, θ_] := r {Sin[θ], Cos[θ]};
  GVector[tail_, tip_, label_, posLabel_, size_, color_] :=
   {color, Arrowheads[size], Arrow[{tail, tip}], Text[Stl[label], posLabel]};

  support = {BLACK, Line[{{-0.5, 0}, {0.5, 0}}]};
  vertical = {GrayLevel[0.75], Dashing[{0.01, 0.01}], Line[{{0, 0}, {0, -1.1}}]};
  {ℓ1, ℓ2} = {μℓ , 1 - μℓ} ℓ;
  {m1, m2} = {μm , 1 - μm} m;
  x1 = ℓ1 Sin[θ1];
  y1 = -ℓ1 Cos[θ1];
  x2 = ℓ1 Sin[θ1] + ℓ2 Sin[θ2];
  y2 = -ℓ1 Cos[θ1] - ℓ2 Cos[θ2];
  {pSize1, pSize2} = {μm, 1 - μm} pSize;
  {O, P1, P2} = {{0, 0}, {x1, y1}, {x2, y2}};

  pendula =
   {Thick, {Brown, Line[{O, P1}], {Black, Text[Stl["ℓ1"], (O + P1)/2 + {1, 0} offset],
      PointSize@pSize1, Point[P1], Text[Stl["m1"], P1 + {1, 0} offset]}},
    {Yellow, Line[{P1, P2}], {Black, Text[Stl["ℓ2"], (P1 + P2)/2 + {1, 0} offset],
      PointSize@pSize2, Point[P2], Text[Stl["m2"], P2 + {1, 0} offset]}}};
  insert = LGrid[{{"θ1", NF3@N[θ1]}, {"ℓ1", NF3@N[ℓ1]}, {"m1", NF3@N[m1]},
     {"θ2", NF3@N[θ2]}, {"ℓ2", NF3@N[ℓ2]}, {"m2", NF3@N[m2]}}, "Parameters"];
  inset = Inset[insert, Scaled[{0.2, 0.8}]];
  Graphics[{support, vertical, pendula, inset},
   PlotLabel → Style[lab, 10, Bold], AspectRatio → 1, ImageSize → {400, 350}]]
```

In[8]:=

```
Clear[ShowEigenmodes];
ShowEigenmodes[{θ1_, θ2_}, μℓ_, μm_, lab_, imageSize_] :=
 Module[{ℓ = 1, m = 1, g = 9.8, offset = 0.075, pSize = 0.10, x1, x2, y1, y2, ℓ1, ℓ2,
   m1, m2, pSize1, pSize2, O, P1, P2, pendula, support, vertical, PtoC, GVector},
  PtoC[r_, θ_] := r {Sin[θ], Cos[θ]};
  GVector[tail_, tip_, label_, posLabel_, size_, color_] :=
   {color, Arrowheads[size], Arrow[{tail, tip}], Text[Stl[label], posLabel]};

  support = {BLACK, Line[{{-0.5, 0}, {0.5, 0}}]};
  vertical = {GrayLevel[0.75], Dashing[{0.01, 0.01}], Line[{{0, 0}, {0, -1.1}}]};
  {ℓ1, ℓ2} = {μℓ , 1 - μℓ} ℓ;
  {m1, m2} = {μm , 1 - μm} m;
  x1 = ℓ1 Sin[θ1];
  y1 = -ℓ1 Cos[θ1];
  x2 = ℓ1 Sin[θ1] + ℓ2 Sin[θ2];
  y2 = -ℓ1 Cos[θ1] - ℓ2 Cos[θ2];
  {pSize1, pSize2} = {μm, 1 - μm} pSize;
  {O, P1, P2} = {{0, 0}, {x1, y1}, {x2, y2}};

  pendula =
   {Thick, {Brown, Line[{O, P1}], {Black, Text[Stl["ℓ₁"], (O + P1)/2 + {1, 0} offset],
      PointSize@pSize1, Point[P1], Text[Stl["m₁"], P1 + {1, 0} offset]}},
    {Yellow, Line[{P1, P2}], {Black, Text[Stl["ℓ₂"], (P1 + P2)/2 + {1, 0} offset],
      PointSize@pSize2, Point[P2], Text[Stl["m₂"], P2 + {1, 0} offset]}}};
  Graphics[{support, vertical, pendula}, PlotLabel → Style[lab, 10, Bold],
   AspectRatio → 1, ImageSize → imageSize]]
```